Diagonalización de matrices, Semejanza.

Practica 8, 28 de abril de 2004

En esta práctica aprenderemos/repasaremos los comandos:

Eigenvalues[matriz] Calcula los autovalores de una matriz
Eigenvectors[matriz] Calcula los autovectores de una matriz
NullSpace[matriz] Calcula una base del Núcleo de unamatriz
DiagonalMatrix[lista] Genera una matriz diagonal cuya diagonal tiene los valores de una lista
MatrixPower[matriz , n] Eleva la matriz a la potencia n
Timing[proceso] Realiza el proceso y devuelve el tiempo usado junto al resultado obtenido

Introducción

Si A es una matriz real cuadrada de dimensión n, el problema de la diagonalización de A consiste en:
−hallar una matriz regular P;
−y hallar una matriz diagonal D tal que:

$$A=P\,D\,P^{-1} \quad \text{o equivalentemente} \quad D=P\,A\,P^{-1}.$$

NOTA:
Cuando dos matrices cuadradas A y D cualesquiera verifican una relación del tipo anterior, diremos que ambas matrices son semejantes.

Supuesto que A sea diagonalizable, veamos quienes han de ser estas matrices P y D.

Escribamos P fijándonos en sus columnas, es decir P = $\begin{pmatrix} x_1 & x_{2.} & \cdots & x_n \end{pmatrix}$ siendo x_i la i−ésimacolumna de P,

y llamemos $\lambda_1, \lambda_2, \ldots, \lambda_n$ a los números que aparecen en la diagonal de la matriz D.
Entonces, reescribiendo la relación entre A, P y D de la siguiente forma

$$P^{-1}\,A\,P = D \;\Rightarrow\; AP = P\,D,$$

y fijándonos en la i−ésimacolumna de las dos igualdades obtenemos:

$$\mathbf{A}\,x_i = \lambda_i\,x_i,$$

para cada i desde 1 hasta n. Por lo tanto, buscar P y D es lo mismo que buscar números λ y vectores x que

cumplan
esta relación.

Cuando se encuentra un **número real** λ y un **vector no nulo** x que verifican la relación $A\,x = \lambda\,x$, diremos que:

$-\lambda$ es un **valor propio** o autovalor de la matriz A;

$-x$ es un **vector propio** o autovector de A asociado al valor propio λ.

Por lo tanto, encontrar la matriz D equivale a encontrar n valores propios (no necesariamente distintos), que serán los que conformen la diagonal de D, y encontrar la matriz P equivale a encontrar n vectores propios **linealmente independientes** (para que P sea regular) asociados a sus respectivos valores propios que conformarán las columnas de P. Así, las columnas de P formarían una base de R^n compuesta únicamente por vectores propios de la matriz A.
Es fundamental, pues, hallar los valores propios de A y los vectores propios asociados. Como los primeros hacen que el sistema $A\,x = \lambda\,x$
tenga solución x distinta de la solución cero, la matriz de coeficientes $A - \lambda\,I$ (donde I denota la matriz identidad de orden n) debe tener determinante no nulo.
Este determinante det(A−λI) **es un polinomio** en λ de grado n y se denomina **polinomio característico** de A.
Por lo tanto los valores propios de A serán los ceros del polinomio característico de A al que se suele notar como

$$p_A(\lambda) = \det(A - \lambda\,I).$$

Por otro lado, el conjunto de vectores propios de A asociados a un mismo valor propio λ forman un subespacio vectorial de R^n que se llama **subespacio propio asociado al valor propio** λ.

Para concluir si una matriz A es o no diagonalizable bastará pues averiguar si hay "suficientes" valores propios para construir D y si hay "suficientes" vectores propios linealmente independientes asociados; esta información nos la dará la dimensión de los subespacios propios y queda recogida en el siguiente resultado.
El **teorema fundamental de la diagonalización** asegura que:
"U*na matriz real cuadrada de orden n es diagonalizable si y sólo si tiene n valores propios reales y la multiplicidad de cada uno de ellos (como raíces del polinomio característico) coincide con la dimensión del subespacio propio asociado*".

Además, el **teorema espectral** nos confirma un caso en el que siempre es posible diagonalizar:
"*Toda matriz real* **simétrica** *es diagonalizable*".
En este caso, se puede conseguir además que las columnas de la matriz de paso P sean una base ortonormal de R^n y, por lo tanto, que P sea una matriz ortogonal, de manera que:

$$\mathrm{A}=P\,\mathrm{D}\,P^{-1} = P\,\mathrm{D}\,P^t\,.$$

Ejemplo 1, matriz diagonalizable.

Veamos cómo se pueden usar las órdenes de Mathematica para estudiar si una matriz es diagonalizable.
Primero, los comandos Eigenvalues y Eigenvectors nos proporcionan respectivamente los valores propios y los vectores propios asociados de dicha matriz.

In[32]:=
```
A={{1, 1, -2}, {6, 9, 1}, {0., -3, -5}};
A//MatrixForm
```

Out[33]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & -2 \\ 6 & 9 & 1 \\ 0. & -3 & -5 \end{pmatrix}$$

In[34]:=
```
pA[λ_]:=Det[A-λ*IdentityMatrix[3]]
pA[λ]            (*también calculamos los valores p usando el Polinomio cco.*
Solve[pA[λ]==0,λ]
valpro=Eigenvalues[A]
```

Out[35]= $24. + 44.\,\lambda + 5\,\lambda^2 - \lambda^3$

Out[36]= {{λ → -4.170476932}, {λ → -0.5896189515}, {λ → 9.760095883}}

Out[37]= {9.760095883, -4.170476932, -0.5896189515}

In[38]:=
```
vectpro=Eigenvectors[A]
```

Out[38]= {{-0.1554285742, -0.968053917, 0.1967576481},
{0.3906460159, -0.2453308117, 0.8872477011},
{0.7753633296, -0.5221650217, 0.3551836106}}

Como todos los valores propios son distintos, todos los subespacios propios son de dimensión uno y por lo tanto la matriz es diagonalizable. Directamente, como nos han dado 3 autovectores, también sabemos que es diagonalizable.
Aunque la orden **Eigenvectors[A]** determina los vectores propios de A, vamos a calcularlos secuencialmente, hallando los subespacios propios uno a uno mediante la orden **NullSpace[ ]**, que calcula una base del núcleo de una matriz dada. Como el subespacio propio asociado a cada valor propio λ viene dado por el núcleo de la matriz $(A - \lambda\, I)$, procederemos uno a uno e iremos añadiendo (por columnas) estos vectores a la matriz P, como lo haríamos a mano.

In[39]:=
```
{vector1}=NullSpace[A-valpro[[1]]*IdentityMatrix[3]]
```

Out[39]= {{0.1554285742, 0.968053917, -0.1967576481}}

In[40]:=
```
{vector2} = NullSpace[A - valpro[[2]] * IdentityMatrix[3]]
```

Out[40]= {{0.3906460159, -0.2453308117, 0.8872477011}}

In[41]:=
```
{vector3} = NullSpace[A - valpro[[3]] * IdentityMatrix[3]]
```

Out[41]= {{-0.7753633296, 0.5221650217, -0.3551836106}}

In[42]:=
```
P={vector1};                    (*iniciamos P con el primer vector*)
P=AppendTo[P,vector2];          (*añadimos el segundo*)
P=AppendTo[P,vector3];          (*añadimos el tercero*)
P=Transpose[P];                 (*trasponemos para que queden por columnas*)
MatrixForm[P]
```

Out[46]//MatrixForm=

$$\begin{pmatrix} 0.1554285742 & 0.3906460159 & -0.7753633296 \\ 0.968053917 & -0.2453308117 & 0.5221650217 \\ -0.1967576481 & 0.8872477011 & -0.3551836106 \end{pmatrix}$$

La matriz diagonal se forma fácilmente con el comando **DiagonalMatrix[lista]**

In[47]:=
```
diag=DiagonalMatrix[valpro];
diag//MatrixForm
```

Out[48]//MatrixForm=

$$\begin{pmatrix} 9.760095883 & 0 & 0 \\ 0 & -4.170476932 & 0 \\ 0 & 0 & -0.5896189515 \end{pmatrix}$$

Es inmediato comprobar los resultados encontrados viendo que A=P D P^{-1}.

In[18]:=
```
A−P.diag.Inverse[P]//MatrixForm
```

Out[18]//MatrixForm=

$$\begin{pmatrix} 6.661338148\times 10^{-16} & 1.110223025\times 10^{-16} & 1.776356839\times 10^{-15} \\ 8.881784197\times 10^{-16} & 0. & 0. \\ 5.4399844\times 10^{-17} & -1.33226763\times 10^{-15} & 0. \end{pmatrix}$$

Intenta explicar este "aparente" error.

Ejemplo 2, matriz no diagonalizable.

Consideramos, ahora, la matriz A, a la que le calculamos los valores propios.

In[19]:=
```
A={{0,1,0},{0,0,1},{2,−5,4}};
Eigenvalues[A]
```

Out[20]= {1, 1, 2}

El valor propio 1 es doble y el valor propio 2 es simple. Para que sea diagonalizable, el subespacio propio asociado al valor propio 1 tendrá que tener dimensión 2.

In[21]:=
```
NullSpace[A-1*IdentityMatrix[3]]
```

Out[21]=
```
{{1, 1, 1}}
```

Como es un subespacio de dimensión uno y el valor propio 1 es doble, la matriz A no es diagonalizable.

Ejemplo 3, matriz simétrica.

Veamos ahora cómo diagonalizar una matriz simétrica (y por lo tanto diagonalizable) de manera que la matriz de paso resulte ortogonal.

In[62]:=
```
A={{-1,1,0},{1,-1,0},{0,0,-2}};
A//MatrixForm
```

Out[63]//MatrixForm=

$$\begin{pmatrix} -1 & 1 & 0 \\ 1 & -1 & 0 \\ 0 & 0 & -2 \end{pmatrix}$$

In[64]:=
```
valpro = Eigenvalues[A]
```

Out[64]=
```
{-2, -2, 0}
```

En este caso, el valor propio $\lambda = -2$ es doble y (como sabemos que A es diagonalizable, por ser simétrica) el subespacio propio tiene dimensión 2. Efectivamente:

In[52]:=
```
{u,v}=NullSpace[A-(-2)*IdentityMatrix[3]]
```

Out[52]=
```
{{0, 0, 1}, {-1, 1, 0}}
```

Para que P resulte ortogonal necesitamos una base ortonormal de este subespacio propio. En este caso, ya son perpendiculares, por lo que únicamente hay que normalizar (si no lo fuesen podemos usar el método de Gram Schmidt):

In[53]:=
```
u = u / Sqrt[u.u]
v = v / Sqrt[v.v]
```

Out[53]= $\{0, 0, 1\}$

Out[54]= $\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right\}$

Por último, añadimos el vector propio asociado al valor propio $\lambda = 0$, (es decir, una base del núcleo de A). Como sabemos, vectores propios asociados a valores propios diferentes siempre son linealmente independientes y además, en el caso de matrices simétricas, son perpendiculares, por lo que bastará normalizar este tercer vector propio para tener construida la base ortonormal y, por lo tanto, la matriz P.

In[55]:=
```
P={u,v};
P=AppendTo[P,NullSpace[A][[1]]];
P[[3]]=P[[3]]/Sqrt[P[[3]].P[[3]]]
P=Transpose[P];
P//MatrixForm
```

Out[57]= $\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right\}$

Out[59]//MatrixForm=

$$\begin{pmatrix} 0 & -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ 0 & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ 1 & 0 & 0 \end{pmatrix}$$

Construímos la matriz diagonal respetando el mismo orden de valores propios $\{-2,-2,0\}$:

In[65]:=
```
diag=DiagonalMatrix[valpro];
diag//MatrixForm
```

Out[66]//MatrixForm=

$$\begin{pmatrix} -2 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Por último, comprobamos tanto que P resulta ortogonal, como la relación de Semejanza:

In[67]:=
```
P.Transpose[P]==IdentityMatrix[3]
A==P.diag.Transpose[P]
```

Out[67]=
```
True
```

Out[68]=
```
True
```

Aplicación: potencias de una matriz.

En ocasiones es necesario calcular potencias "elevadas" de una matriz dada o incluso "raices". En estos casos es útil disponer de una matriz semejante a la dada a la que sea fácil calcularle esas potencias o ríces para disminuir el coste computacional. La simplificación se basa en que:

$$\mathrm{A}=P\,\mathrm{D}\,P^{-1} \quad \Rightarrow \quad A^m = P\,D^m\,P^{-1},$$

Si la matriz D es diagonal, sus potencias se calculan simplemente haciendo la misma potencia (incluso fraccionaria) de cada uno de los elementos de la diagonal, lo que conlleva menos cómputo que hacerlo para una matriz arbitraria.

Ejercicio: Para la sigiente matriz

$$A = \begin{pmatrix} 1 & 2 & \ldots & 10 \\ 2 & 3 & \ldots & 11 \\ \vdots & \vdots & \ddots & \vdots \\ 10 & 11 & \ldots & 19 \end{pmatrix},$$

calculemos por un lado A^{8000} y por otro, calculemos una matriz B tal que $B^3 = A$. A la matriz B se le llama "una raíz cúbica de la matriz A".

In[69]:=
```
A = Table[i + j - 1, {i, 10}, {j, 10}];
MatrixForm[A]
```

Out[70]//MatrixForm=

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 \\ 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 & 11 \\ 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 & 11 & 12 \\ 4 & 5 & 6 & 7 & 8 & 9 & 10 & 11 & 12 & 13 \\ 5 & 6 & 7 & 8 & 9 & 10 & 11 & 12 & 13 & 14 \\ 6 & 7 & 8 & 9 & 10 & 11 & 12 & 13 & 14 & 15 \\ 7 & 8 & 9 & 10 & 11 & 12 & 13 & 14 & 15 & 16 \\ 8 & 9 & 10 & 11 & 12 & 13 & 14 & 15 & 16 & 17 \\ 9 & 10 & 11 & 12 & 13 & 14 & 15 & 16 & 17 & 18 \\ 10 & 11 & 12 & 13 & 14 & 15 & 16 & 17 & 18 & 19 \end{pmatrix}$$

Como la matriz es simétrica, sabemos que es diagonalizable. La diagonalizamos.

In[79]:=
```
valpro=Eigenvalues[A]
diag=DiagonalMatrix[valpro];
```

Out[79]= $\{0, 0, 0, 0, 0, 0, 0, 0, 5\left(10-\sqrt{133}\right), 5\left(10+\sqrt{133}\right)\}$

Dado que el conjunto de valores propios sigue siendo sencillo (0 con multiplicidad 8 y dos valores más que ocupan la novena y décima posición), repetimos el proceso anterior (hágase también usando la orden **Eigenvectors**).

In[72]:=
```
i=IdentityMatrix[10]//N;
P=NullSpace[A];
P=AppendTo[P,NullSpace[A-valpro[[9]]*i][[1]]];
P=AppendTo[P,NullSpace[A-valpro[[10]]*i][[1]]];
P=Transpose[P];
MatrixForm[P]
```

Out[77]//MatrixForm=

$$\begin{pmatrix}
8 & 7 & 6 & 5 & 4 & 3 & 2 & 1 & 0.5602050635 & -0.1778337207 \\
-9 & -8 & -7 & -6 & -5 & -4 & -3 & -2 & 0.4538292298 & -0.2062131873 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0.3474533961 & -0.2345926539 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0.2410775624 & -0.2629721205 \\
0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0.1347017287 & -0.2913515871 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0.028325895 & -0.3197310537 \\
0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & -0.0780499387 & -0.3481105203 \\
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & -0.1844257724 & -0.3764899869 \\
0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & -0.2908016061 & -0.4048694535 \\
1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -0.3971774398 & -0.4332489201
\end{pmatrix}$$

Comprobamos la semejanza y calculamos las potencias pedidas.
Puesto que el penúltimo valor propio es negativo, para evitar una raíz cúbica compleja lo hacemos positivo y luego le volvemos a colocar el signo negativo en su sitio.

In[81]:=
```
A==P.diag.Inverse[P]
```

Out[81]= True

In[82]:=
```
valpro8000 = N[valpro]^8000;
valprocubic = Sign[valpro] N[Abs[valpro]]^(1/3);
diag8000 = DiagonalMatrix[valpro8000];
diagcubic = DiagonalMatrix[valprocubic];
```

In[86]:=
```
B = P.diagcubic.Inverse[P];
MatrixForm[B]
```

Out[87]//MatrixForm=

-0.4682672947	-0.3267723313	-0.1852773679	-0.0437824045	0.09771255891
-0.3267723313	-0.2037550121	-0.08073769287	0.04227962635	0.1652969456
-0.1852773679	-0.08073769287	0.02380198216	0.1283416572	0.2328813322
-0.0437824045	0.04227962635	0.1283416572	0.2144036881	0.3004657189
0.09771255891	0.1652969456	0.2328813322	0.3004657189	0.3680501056
0.2392075223	0.2883142648	0.3374210073	0.3865277498	0.4356344922
0.3807024857	0.411331584	0.4419606823	0.4725897806	0.5032188789
0.5221974491	0.5343489032	0.5465003574	0.5586518115	0.5708032656
0.6636924125	0.6573662225	0.6510400324	0.6447138423	0.6383876522
0.805187376	0.7803835417	0.7555797074	0.7307758732	0.7059720389

In[88]:=
```
P.diag8000 .Inverse[P] // MatrixForm
```

Out[88]//MatrixForm=

$1.061230962260 \times 10^{16255}$	$1.23058674313 \times 10^{16255}$	$1.39994252400 \times 10^{16255}$	1.569298
$1.23058674313 \times 10^{16255}$	$1.42696904465 \times 10^{16255}$	$1.62335134617 \times 10^{16255}$	1.819733
$1.39994252400 \times 10^{16255}$	$1.62335134617 \times 10^{16255}$	$1.84676016833 \times 10^{16255}$	2.070168
$1.56929830487 \times 10^{16255}$	$1.81973364768 \times 10^{16255}$	$2.07016899050 \times 10^{16255}$	2.320604
$1.73865408574 \times 10^{16255}$	$2.01611594920 \times 10^{16255}$	$2.29357781267 \times 10^{16255}$	2.571039
$1.90800986661 \times 10^{16255}$	$2.21249825072 \times 10^{16255}$	$2.51698663483 \times 10^{16255}$	2.821475
$2.07736564748 \times 10^{16255}$	$2.40888055224 \times 10^{16255}$	$2.74039545700 \times 10^{16255}$	3.071910
$2.24672142835 \times 10^{16255}$	$2.60526285376 \times 10^{16255}$	$2.96380427916 \times 10^{16255}$	3.322345
$2.41607720922 \times 10^{16255}$	$2.80164515527 \times 10^{16255}$	$3.18721310133 \times 10^{16255}$	3.572781
$2.58543299009 \times 10^{16255}$	$2.99802745679 \times 10^{16255}$	$3.41062192350 \times 10^{16255}$	3.823216

Verificamos la construcción de B.

In[89]:=
```
MatrixPower[B, 3] == A
```

Out[89]=
```
True
```

También comprobamos (en esta matriz 10 x 10) la diferencia de tiempo si usamos la ordenMatrixPower (El comando **Timing** produce el tiempo de ejecución).

In[90]:=
```
MatrixPower[A, 8000] // N // MatrixForm // Timing
```

Out[90]=
$$\{4.43\ \text{Second},$$

$1.061230962260368 \times 10^{16255}$	$1.230586743130217 \times 10^{16255}$	$1.399942524000066 \times 10^{16}$
$1.230586743130217 \times 10^{16255}$	$1.426969044648264 \times 10^{16255}$	$1.623351346166310 \times 10^{16}$
$1.399942524000066 \times 10^{16255}$	$1.623351346166310 \times 10^{16255}$	$1.846760168332555 \times 10^{16}$
$1.569298304869915 \times 10^{16255}$	$1.819733647684357 \times 10^{16255}$	$2.070168990498799 \times 10^{16}$
$1.738654085739765 \times 10^{16255}$	$2.016115949202404 \times 10^{16255}$	$2.293577812665043 \times 10^{16}$
$1.908009866609614 \times 10^{16255}$	$2.212498250720451 \times 10^{16255}$	$2.516986634831287 \times 10^{16}$
$2.077365647479463 \times 10^{16255}$	$2.408880552238497 \times 10^{16255}$	$2.740395456997532 \times 10^{16}$
$2.246721428349312 \times 10^{16255}$	$2.605262853756544 \times 10^{16255}$	$2.963804279163776 \times 10^{16}$
$2.416077209219161 \times 10^{16255}$	$2.801645155274591 \times 10^{16255}$	$3.187213101330020 \times 10^{16}$
$2.585432990089010 \times 10^{16255}$	$2.998027456792637 \times 10^{16255}$	$3.410621923496264 \times 10^{16}$

Ejercicios

1.−Diagonaliza la siguiente matriz: $A=\begin{pmatrix} 5 & 2 & 1 \\ 2 & 7 & 0 \\ 1 & 0 & -3 \end{pmatrix}.$

2.−Verifica, con los datos obtenidos en el ejercicio anterior, la relación:

$$A^5 = P\,D^5\,P^{-1}.$$

3.−Calcula los valores propios de la matriz $A=\begin{pmatrix} 2 & 5 & 0 \\ 0 & 3 & 5 \\ 0 & 0 & 3 \end{pmatrix}$ y las dimensiones de los subespacios propios asociados a cada uno.

Determina si es o no diagonalizable.